ZASTOSOWANIA MATEMATYKI
APPLICATIONES MATHEMATICAE
XVI, 4 (1980)

ALGORITHMS 70-71

E. NEUMAN (Wrocław)

DETERMINATION OF A QUADRATIC SPLINE FUNCTION WITH GIVEN VALUES OF THE INTEGRALS IN SUBINTERVALS

0. Introduction. Descriptions of two procedures *integr2s0sncf* and *integr2s10s1ncf* which determine the quadratic spline functions with given values of the integrals in subintervals and satisfying some boundary conditions are given.

1. Procedure declaration. Let n be a natural number ($n > 1$). For given knots x_j ($j = 0, 1, \ldots, n$), where $x_j < x_{j+1}$ for all $j = 0, 1, \ldots, n-1$, and given real numbers f_j ($j = 1, 2, \ldots, n$) we determine a quadratic spline function s such that

(i) $s \in C^1[x_0, x_n]$;

(ii) in each subinterval $[x_j, x_{j+1}]$ ($j = 0, 1, \ldots, n-1$), s is an algebraic polynomial of degree at most 2;

(iii) $\int_{x_{j-1}}^{x_j} s(x)\,dx = f_j$ ($j = 1, 2, \ldots, n$).

Additionally, the function s determined by conditions (i)-(iii) satisfies one of the following boundary conditions:

(iv) $s(x_0) = s_0$, $s(x_n) = s_n$ in the case of procedure *integr2s0sncf*,

(iv′) $s'(x_0) = s_0'$, $s'(x_n) = s_n'$ in the case of procedure *integr2s10s1ncf*,

where s_0, s_n, s_0', s_n' are given real numbers.

Data:

n — number of knots of the spline function s minus one;

$x[0:n]$ — array of knots of the spline function s;

$f[1:n]$ — array of integrals of the spline functions (see conditions (iii));

s0, *sn* — values of the spline function s at knots x_0 and x_n, respectively (in the case of procedure *integr2s0sncf* only);

s10, *s1n* — values of the first derivative of the spline function s at knots x_0 and x_n, respectively (in the case of procedure *integr2s10s1ncf* only).

```
procedure integr2s0sncf(n,x,f,s0,sn,a,b,c,h);
 value n;
 integer n;
 real s0,sn;
 array x,f,a,b,c,h;
 begin
  integer j,j1,n1;
  real fj,fj1,s,s1,xj,xj1;
  array q,u[1:n-1];
  xj:=x[0];
  for j:=1 step 1 until n do
   begin
    xj1:=x[j];
    h[j]:=xj1-xj;
    xj:=xj1
   end j;
  n1:=n-1;
  xj:=h[1];
  fj:=f[1];
  for j:=1 step 1 until n1 do
   begin
    j1:=j+1;
    xj1:=h[j1];
    fj1:=f[j1];
    s:=b[j]:=xj1/(xj+xj1);
    s1:=c[j]:=1.0-s;
    a[j]:=3.0×(fj×s/xj+fj1×s1/xj1);
    xj:=xj1;
    fj:=fj1
   end j;
```

```
s:=a[1]:=a[1]-s0×b[1];
a[n1]:=a[n1]-sn×c[n1];
fj:=q[1]:=-.5×c[1];
xj:=u[1]:=.5×s;
for j:=2 step 1 until n1 do
 begin
  xj1:=b[j];
  s:=xj1×fj+2.0;
  fj:=q[j]:=-c[j]/s;
  xj:=u[j]:=(a[j]-xj1×xj)/s
 end j;
c[0]:=s0;
c[n]:=sn;
c[n1]:=xj;
for j:=n-2 step -1 until 1 do
 xj:=c[j]:=q[j]×xj+u[j];
xj:=s0;
for j:=0 step 1 until n1 do
 begin
  j1:=j+1;
  xj1:=c[j1];
  fj:=h[j1];
  s:=a[j]:=3.0×(xj+xj1-2.0×f[j1]/fj)/(fj×fj);
  b[j]:=((xj1-xj)/fj-fj×s);
  xj:=xj1
 end j
end integr2s0sncf
```

```
procedure integr2s10s1ncf(n,x,f,s10,s1n,a,b,c,h);
 value n,s10,s1n;
 integer n;
 real s10,s1n;
 array x,f,a,b,c,h;
 begin
  integer j,j1,n1;
  real fj,fj1,xj,xj1,s;
  array q,u[1:n-1];
  xj:=x[0];
  for j:=1 step 1 until n do
   begin
    xj1:=x[j];
    h[j]:=xj1-xj;
    xj:=xj1
   end j;
  n1:=n-1;
  xj:=h[1];
  fj:=f[1];
  for j:=1 step 1 until n1 do
   begin
    j1:=j+1;
    xj1:=h[j1];
    fj1:=f[j1];
    s:=a[j]:=xj/(xj+xj1);
    c[j]:=1.0-s;
    b[j]:=6.0×(fj1/xj1-fj/xj)/(xj+xj1);
    xj:=xj1;
    fj:=fj1
   end j;
```

```
s:=b[1]:=b[1]-a[1]×s10;
b[n1]:=b[n1]-c[n1]×s1n;
fj:=q[1]:=-.5×c[1];
s:=u[1]:=.5×s;
for j:=2 step 1 until n1 do
 begin
  xj:=a[j];
  xj1:=2.0+xj×fj;
  fj:=q[j]:=-c[j]/xj1;
  s:=u[j]:=(b[j]-xj×s)/xj1
 end j;
fj:=b[0]:=s10;
b[n]:=s1n;
b[n1]:=s;
for j:=n-2 step -1 until 1 do
 s:=b[j]:=q[j]×s+u[j];
for j:=0 step 1 until n1 do
 begin
  j1:=j+1;
  fj1:=b[j1];
  xj:=h[j1];
  a[j]:=(fj1-fj)/(2.0×xj);
  c[j]:=f[j1]/xj-.1666666667×xj×(2.0×fj+fj1);
  fj:=fj1
 end j
end integr2s10s1ncf
```

Results:

$a[0:n-1]$, $b[0:p]$, $c[0:q]$ — arrays of coefficients of the spline function s, where

$$p = \begin{cases} n-1 & \text{in the case of procedure } \textit{integr2s0sncf}, \\ n & \text{in the case of procedure } \textit{integr2s10s1ncf}, \end{cases}$$

$$q = \begin{cases} n & \text{in the case of procedure } \textit{integr2s0sncf}, \\ n-1 & \text{in the case of procedure } \textit{integr2s10s1ncf}; \end{cases}$$

$h[1:n]$ — array of mesh sizes ($h_j = x_j - x_{j-1}$; $j = 1, 2, \ldots, n$).

In each subinterval $[x_j, x_{j+1}]$ ($j = 0, 1, \ldots, n-1$) the spline function s is of the form

$$s(x) = a_j t^2 + b_j t + c_j, \quad \text{where } t = x - x_j.$$

2. Methods used. We begin with the description of the method used for determining the spline function s described by conditions (i)-(iv). Let $s_j = s(x_j)$ ($j = 0, 1, \ldots, n$). From (2.8) in [1] it follows that the numbers s_j must satisfy the system of linear equations

$$(1) \qquad \lambda_j s_{j-1} + 2s_j + \mu_j s_{j+1} = 3\left(\frac{\lambda_j f_j}{h_j} + \frac{\mu_j f_{j+1}}{h_{j+1}}\right) \quad (j = 1, 2, \ldots, n-1)$$

(s_0 and s_n are given), where $h_j = x_j - x_{j-1}$ ($j = 1, 2, \ldots, n-1$), and

$$(2) \qquad \lambda_j = \frac{h_{j+1}}{h_j + h_{j+1}}, \quad \mu_j = 1 - \lambda_j \quad (j = 1, 2, \ldots, n-1).$$

For $x_j \leqslant x \leqslant x_{j+1}$ the spline function $s(x)$ takes the form

$$(3) \qquad s(x) = \frac{x - x_j}{h_j} s_{j+1} + \frac{x_{j+1} - x}{h_{j+1}} s_j + d_{j+1}(x - x_j)(x_{j+1} - x),$$

where the numbers d_{j+1} are equal to

$$d_{j+1} = \frac{3}{h_{j+1}^2}\left(2\frac{f_{j+1}}{h_{j+1}} - s_j - s_{j+1}\right) \quad (j = 0, 1, \ldots, n-1)$$

(see equation (2.7) in [1]). Hence and from (3) it follows that, for $x_j \leqslant x \leqslant x_{j+1}$,

$$s(x) = a_j t^2 + b_j t + c_j,$$

where

$$a_j = -d_{j+1}, \quad b_j = \frac{s_{j+1} - s_j}{h_{j+1}} + d_{j+1} h_{j+1}, \quad c_j = s_j, \quad t = x - x_j$$
$$(j = 0, 1, \ldots, n-1).$$

Now we present the method used for determining the spline function s for which conditions (i)-(iii) and (iv′) hold. Let $m_j = s'(x_j)$ ($j = 0, 1, \ldots, n$). From equation (2.5) in [1] it follows that the numbers m_j must satisfy the system of linear equations

$$(4) \quad \mu_j m_{j-1} + 2m_j + \lambda_j m_{j+1} = \frac{6}{h_j + h_{j+1}}\left(\frac{f_{j+1}}{h_{j+1}} - \frac{f_j}{h_j}\right) \quad (j = 1, 2, \ldots, n-1)$$

(here m_0 and m_n are given), where μ_j and λ_j are the same as in (2). For $x_j \leqslant x \leqslant x_{j+1}$ the spline function s takes the form

$$(5)\quad s(x) = -\frac{(x_{j+1}-x)^2}{2h_{j+1}} m_j + \frac{(x-x_j)^2}{2h_{j+1}} m_{j+1} + e_{j+1} \qquad (j = 0, 1, \ldots, n-1),$$

where the numbers e_{j+1} are equal to

$$e_{j+1} = \frac{f_{j+1}}{h_{j+1}} + \frac{h_{j+1}}{6}(m_j - m_{j+1}) \qquad (j = 0, 1, \ldots, n-1).$$

Hence and from (5) it follows that, for $x_j \leqslant x \leqslant x_{j+1}$,

$$s(x) = a_j t^2 + b_j t + c_j,$$

where

$$a_j = \frac{m_{j+1}-m_j}{2h_{j+1}}, \quad b_j = m_j, \quad c_j = \frac{f_{j+1}}{h_{j+1}} - \frac{h_{j+1}}{6}(2m_j + m_{j+1}), \quad t = x - x_j$$
$$(j = 0, 1, \ldots, n-1).$$

The systems of linear equations (1) and (4) may be solved by the LU decomposition (matrices of the above systems are diagonally dominant).

3. Certification. Let the quantities $E_{\max}$, E_0 and E_1 be defined as follows:

$$E_{\max} = \max_{1\leqslant j\leqslant n} \left| \int_{x_{j-1}}^{x_j} s(x)\,dx - f_j \right|, \quad E_0 = \max_{1\leqslant j\leqslant n-1} |s(x_j+) - s(x_j-)|,$$
$$E_1 = \max_{1\leqslant j\leqslant n-1} |s'(x_j+) - s'(x_j-)|.$$

In the following examples the values $E_{\max}$, E_0 and E_1 are given for various values of n.

Example 3.1. Let

$$x_j = \frac{j}{n} \qquad (j = 0, 1, \ldots, n),$$

$$f_j = a_j - a_{j-1}, \quad a_j = \frac{x_j^2}{6}(2-3x_j) \quad (j = 1, 2, \ldots, n), \quad s0 = 0, \quad sn = .1.$$

The results of using procedures *integr2s0sncf* and *integr2s10s1ncf* are given in Tables 3.1 and 3.1′, respectively.

TABLE 3.1

n	$E_{\max}$	E_0	E_1
10	$.705_{10}-11$	.0	$.513_{10}-9$
100	$.742_{10}-12$	.0	$.874_{10}-8$
500	$.151_{10}-12$	$.182_{10}-11$	$.415_{10}-7$

TABLE 3.1′

n	$E_{\max}$	E_0	E_1
10	$.909_{10}-12$	$.728_{10}-11$	$.364_{10}-11$
100	$.568_{10}-13$	$.728_{10}-11$	$.114_{10}-12$
500	$.711_{10}-14$	$.364_{10}-11$	.0

Example 3.2. Let

$$x_j = \sin\frac{\pi}{2n}j \quad (j = 0, 1, \ldots, n),$$

$$f_j = \exp(x_j) - \exp(x_{j-1}) \quad (j = 1, 2, \ldots, n), \quad s10 = 0, \quad s1n = .1.$$

The results of using procedures *integr2s0sncf* and *integr2s10s1ncf* are given in Tables 3.2 and 3.2′, respectively.

TABLE 3.2

n	E_{max}	E_0	E_1
10	$.546_{10}-9$	$.146_{10}-10$	$.119_{10}-7$
100	$.573_{10}-10$	.0	$.850_{10}-7$
500	$.108_{10}-10$	$.291_{10}-10$	$.627_{10}-6$

TABLE 3.2′

n	E_{max}	E_0	E_1
10	$.109_{10}-10$	$.582_{10}-10$	.0
100	$.136_{10}-11$	$.116_{10}-9$	.0
500	$.341_{10}-12$	$.116_{10}-9$	.0

All calculations were done on the Odra 1204 computer. The execution time of procedure *integr2s0sncf* is approximately equal to $34n+1010$ msec and that of *integr2s10s1ncf* to $33n+1008$ msec.

Reference

[1] A. Sharma and J. Tzimbalario, *Quadratic splines*, J. Approximation Theory 19 (1977), p. 186-193.

INSTITUTE OF INFORMATICS
UNIVERSITY OF WROCŁAW
50-384 WROCŁAW

Received on 5. 8. 1977

ALGORYTMY 70-71

E. NEUMAN (Wrocław)

WYZNACZANIE FUNKCJI SKLEJANEJ STOPNIA DRUGIEGO Z ZADANYMI WARTOŚCIAMI CAŁEK W PODPRZEDZIAŁACH

STRESZCZENIE

Niech n będzie liczbą naturalną ($n > 1$). Dla danych węzłów x_j ($j = 0, 1, \ldots, n$), gdzie $x_j < x_{j+1}$ dla wszystkich $j = 0, 1, \ldots, n-1$, oraz danych liczb rzeczywistych f_j ($j = 1, 2, \ldots, n$) wyznacza się współczynniki takiej funkcji sklejanej s stopnia drugiego, że

(i) $s \in C^1[x_0, x_n]$;

(ii) w każdym podprzedziale $[x_j, x_{j+1}]$ $(j = 0, 1, \ldots, n-1)$ s jest wielomianem algebraicznym stopnia co najwyżej 2;

(iii) $\int_{x_{j-1}}^{x_j} s(x)dx = f_j$ $(j = 1, 2, \ldots, n)$.

Funkcja s opisana przez (i)-(iii) spełnia dodatkowo jeden z następujących warunków brzegowych:

(iv) $s(x_0) = s_0$, $s(x_n) = s_n$ w przypadku procedury *integr2s0sncf*,

(iv′) $s'(x_0) = s'_0$, $s'(x_n) = s'_n$ w przypadku procedury *integr2s10s1ncf*,

gdzie s_0, s_n, s'_0, s'_n są danymi liczbami rzeczywistymi.

Dane:

n — liczba węzłów funkcji s minus jeden;

x[0 : n] — tablica węzłów funkcji s;

f[1 : n] — tablica wartości całek funkcji s w poszczególnych podprzedziałach (patrz warunki (iii));

s0, sn — wartości funkcji s odpowiednio w węzłach x_0 i x_n (w przypadku procedury *integr2s0sncf*);

s10, s1n — wartości pierwszej pochodnej funkcji s odpowiednio w węzłach x_0 i x_n (w przypadku procedury *integr2s10s1ncf*).

Wyniki:

a[0 : n−1], b[0 : p], c[0 : q] — tablice współczynników funkcji s, gdzie

$$p = \begin{cases} n-1 & \text{w przypadku procedury } \textit{integr2s0sncf}, \\ n & \text{w przypadku procedury } \textit{integr2s10s1ncf}, \end{cases}$$

$$q = \begin{cases} n & \text{w przypadku procedury } \textit{integr2s0sncf}, \\ n-1 & \text{w przypadku procedury } \textit{integr2s10s1ncf}; \end{cases}$$

h[1 : n] — tablica odległości między sąsiednimi węzłami ($h_j = x_j - x_{j-1}$; $j = 1, 2, \ldots, n$).

W każdym podprzedziale $[x_j, x_{j+1}]$ $(j = 0, 1, \ldots, n-1)$ funkcja s ma postać

$$s(x) = a_j t^2 + b_j t + c_j, \quad \text{gdzie } t = x - x_j.$$